

Three Level Models for Longitudinal Twin Data (Time within Twin within Pair)

The data for this example come from the Octogenarian Twin Study of Aging, a longitudinal study. These models include 351 same-sex twin pairs initially age 79–100 years measured for up to four occasions every two years, over six possible years. We will be examining change over time in a measure of crystallized intelligence (information test), as well the extent of heritability (i.e., differences between MZ and DZ twins) in intercepts and change over time. These data are already stacked such that one row contains the data for one occasion for one person. The ID variables PairID and TwinID index which twin pair and which twin (1 or 2), respectively. Time is not balanced across persons, so REPEATED will not be used until we get to the heritability models (i.e., that include different variances by zygosity).

Model 1a: Empty Means, Two-Level Model for Information Test Outcome

Level 1: $Info_{ti} = \beta_{0i} + e_{ti}$ Level 2: $\beta_{0i} = \gamma_{00} + U_{0i}$	This model has two variance components: level-1 residual and level-2 random intercept. It assumes that all people are independent (i.e., it does not account for twin pair membership).
--------------------------------------------------------------------------------------------	-----------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------

```

TITLE "SAS Model 1a: Empty Means, Two-Level Model for Information Test Outcome";
PROC MIXED DATA=work.Example8 NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
CLASS PairID TwinID;
MODEL info = / SOLUTION DDFM=Satterthwaite;
RANDOM INTERCEPT / VCORR TYPE=UN SUBJECT=PairID*TwinID; * Level 2+3 combined;
ODS OUTPUT InfoCrit=Fit2L; * Save fit stats for LRT; RUN;

* STATA Model 1a: Empty Means, Two-Level Model for Information Test Outcome
mixed info , || Case: , variance reml covariance(un) dfmethod(satterthwaite) dftable(pvalue)
estat ic, n(702) // Giving STATA highest-level sample size to use for BIC
estat icc // Requesting intraclass correlation
estimates store TwoLevel
    
```

Case is a person-level ID variable needed just for this model in STATA.

SAS output:

Dimensions					
Covariance Parameters					2
Columns in X					1
Columns in Z Per Subject					1
Subjects					702 → number of persons so far
Max Obs Per Subject					4
Covariance Parameter Estimates					
				Standard	Z
Cov Parm	Subject	Estimate	Error	Value	Pr > Z
UN(1,1)	PairID*TwinID	136.53	8.5293	16.01	<.0001
Residual		23.9167	1.0694	22.36	<.0001
Null Model Likelihood Ratio Test					
DF	Chi-Square	Pr > ChiSq			
1	1333.46	<.0001			
Information Criteria					
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC
11389.5	2	11393.5	11393.5	11397.0	11402.6
Solution for Fixed Effects					
				Standard	
Effect	Estimate	Error	DF	t Value	Pr > t
Intercept	25.5469	0.4911	605	52.02	<.0001

Calculate the ICC for the proportion of between-person variation in Info:

$$ICC = \frac{136.53}{136.53 + 23.92} = .85$$

The “Null Model” LRT below tells us that the random intercept variance is significantly greater than 0, and thus so is the ICC for the correlation of occasions within persons (and pairs).

Model 1b: Empty Means, Three-Level Model for Information Test Outcome

Level 1: $Info_{ij} = \beta_{0ij} + e_{ij}$
 Level 2: $\beta_{0ij} = \delta_{00j} + U_{0ij}$
 Level 3: $\delta_{00j} = \gamma_{000} + V_{00j}$

This model now has 3 variance components: level-1 residual, level-2 twin random intercept, and level-3 pair random intercept. It now allows a correlation between people from the same twin pair.

```
TITLE "SAS Model 1b: Empty Means, Three-Level Model for Information Test Outcome";
PROC MIXED DATA=work.Example8 NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS PairID TwinID;
  MODEL info = / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID;           * Level 3;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID*TwinID;    * Level 2;
  ODS OUTPUT InfoCrit=Fit3L CovParms=CovEmpty; * Save for LRT, Pseudo-R2; RUN;
* Compare three-level empty to two-level empty;
%FitTest(FitFewer=Fit2L, FitMore=Fit3L);

* STATA Model 1b: Empty Means, Three-Level Model for Information Test Outcome
mixed info , || PairID: , covariance(unstructured) ///
             || TwinID: , covariance(unstructured) variance reml ///
             dfmethod(satterthwaite) dftable(pvalue),
estimates store ThreeLevel
lrtest ThreeLevel TwoLevel
```

TwinID is sufficient for level 2 here because STATA assumes any random effects written after the first are nested within the first, whereas SAS does not. I am not requesting ICC from STATA because it gives L3/total instead of L3/L2+L3.

SAS output:

Dimensions						
Covariance Parameters					3	
Columns in X					1	
Columns in Z Per Subject					3	
Subjects					351 → now number of twin pairs (families)	
Max Obs Per Subject					8 → per twin pair (4 occasions * 2 persons)	
Covariance Parameter Estimates						
				Standard	Z	
Cov Parm	Subject	Estimate	Error	Value	Pr > Z	
UN(1,1)	PairID	87.2970	9.9794	8.75	<.0001 → level-3 between-pair	
UN(1,1)	PairID*TwinID	49.9360	5.3371	9.36	<.0001 → level-2 within-pair	
Residual		23.9684	1.0735	22.33	<.0001 → level-1 within-person	
Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
11278.1	3	11284.1	11284.1	11288.7	11295.7	11298.7
Solution for Fixed Effects						
				Standard		
Effect	Estimate	Error	DF	t Value	Pr > t	
Intercept	25.2203	0.6017	331	41.92	<.0001	

Is the 3-level model a better fit than the 2-level model?
 Yes, $-2\Delta LL(\sim 1) = 111.37, p < .001$

Likelihood Ratio Test for Fit2L vs. Fit3L							
	Neg2Log						
Name	Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
Fit2L	11389.5	2	11393.5	11402.6	.	.	.
Fit3L	11278.1	3	11284.1	11295.7	111.373	1	0

Proportion variance at each level:

Total = $87.30 + 49.94 + 23.97 = 161.20$
 Level 3 (pair) = $87.30 / 161.20 = .54$
 Level 2 (person) = $49.94 / 161.20 = .31$
 Level 1 (time) = $23.97 / 161.20 = .15$

ICC_{L2} for time within person and pair =
 $(87.30 + 49.94) / (161.20) = .85$

ICC_{L3} for person within pair = $87.30 / (87.30 + 49.94) = .64$
 This ICC = .64 is significantly greater than 0 via $-2\Delta LL$ for 3- vs. 2-level.

Now let's do the same thing for our time-varying predictor of age:

```
TITLE "SAS Age Model: Empty Means, Three-Level Model for Age Predictor";
PROC MIXED DATA=work.Example8 NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS PairID TwinID;
  MODEL age = / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID;          * Level 3;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID*TwinID;   * Level 2; RUN;

* STATA Age Model: Empty Means, Three-Level Model for Age Predictor
mixed age , || PairID : , covariance(unstructured) ///
           || TwinID : , covariance(unstructured) variance reml ///
           dfmethod(satterthwaite) dftable(pvalue),
```

SAS output:

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	PairID	8.5256	0.7193	11.85	<.0001 level-3 between-pair = 63%
UN(1,1)	PairID*TwinID	0	.	.	. level-2 within-pair = 0%
Residual		4.9682	0.1693	29.35	<.0001 level-1 within-person = 37%

Because there is no age variance at level 2, age will be a predictor at levels 1 and 3 only.

Below we create our predictors: level-1 (time-varying) age will be time-in-study (0=baseline), and level-3 (between-pair) age will be baseline age centered at 85 years. This creates a clear demarcation of age at baseline as the cross-sectional effect of age, and time-in-study as the longitudinal effect of age.

SAS Data Manipulation:

```
DATA work.Example8; SET work.Example8;
* Centering age at time 1 at 85 to use at level 3;
  BFace85 = agew1 - 85; LABEL BFace85= "BFace85: Age at Time1 (0=85)";
* Within-person centering age at level-1 (VARIABLE-BASED CENTERING);
  time = age - agew1; LABEL time= "time: Time Since Entry (0= Age Wave 1)";
* Make string version of zygosity for easier output reading;
  IF zygosity=1 THEN zyg="MZ"; IF zygosity=2 THEN zyg="DZ";
* Selecting only cases with complete data;
  IF NMISS(agew1, age, info)>0 THEN DELETE; RUN;
```

STATA Data Manipulation:

```
* Centering age at time 1 at 85 to use at level 3
gen BFace85 = agew1 - 85
label variable BFace85 "BFace85: Age at Time1 (0=85)"
* Within person centering age at level-1 (VARIABLE-BASED CENTERING)
gen time = age - agew1
label variable time "time: Time since entry (0= Age Wave 1)"
* Recode zygosity so 0=DZ, 1=MZ, will be treated as numeric
gen zyg = zygosity-1
* Selecting only cases with complete data
egen nummiss = rowmiss(agew1 age, info)
drop if nummiss>0
```

Model 1c: Saturated Means for Wave, Random Intercepts at Levels 2 and 3

Using SAS GLIMMIX instead of SAS MIXED to get a means plot directly


```
TITLE "SAS Model 1c: Saturated Wave Means, Three-Level Model for Information Test Outcome";
PROC GLIMMIX DATA=work.Example8a NOCLPRINT NAMELEN=100 METHOD=RSPL; * Same as REML;
  CLASS PairID TwinID Wave;
  MODEL info = Wave / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID;          * Level 3;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID*TwinID;   * Level 2;
  LSMEANS Wave / PLOT=MEANPLOT(CLBAND JOIN); RUN;          * Print and plot means;
```

```
* STATA Model 1c: Saturated Wave Means, Three-Level Model for Information Test Outcome
mixed age i.Wave, || PairID: , covariance(unstructured) ///
|| TwinID: , covariance(unstructured) variance reml ///
dfmethod(satterthwaite) dftable(pvalue),

    margins i.Wave
    marginsplot
```

Wave	Least Squares Estimate	Standard Error
1	26.0881	0.6247
2	25.4596	0.6384
3	23.9172	0.6575
4	22.9877	0.6809

This pattern of average change looks like it might need a fixed quadratic effect of time, so let's start there.



Model 2a: Fixed Quadratic Time, Random Intercepts at Levels 3 (Pair) and 2 (Twin)

Level 1: $Info_{tj} = \beta_{0ij} + \beta_{1ij}(Age_{tj} - PairAge1_j) + \beta_{2ij}(Age_{tj} - PairAge1_j)^2 + e_{tj}$

Level 2:

- Intercept: $\beta_{0ij} = \delta_{00j} + U_{0ij}$
- Linear Time: $\beta_{1ij} = \delta_{10j}$
- Quadratic Time: $\beta_{2ij} = \delta_{20j}$

Level 3:

- Intercept: $\delta_{00j} = \gamma_{000} + V_{00j}$
- Linear Time: $\delta_{10j} = \gamma_{100}$
- Quadratic Time: $\delta_{20j} = \gamma_{200}$

```
TITLE "SAS Model 2a: Fixed Quadratic Time, Random Intercepts for Pair and Twin";
PROC MIXED DATA=work.Example8a NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
CLASS PairID TwinID;
MODEL info = time time*time / SOLUTION DDFM=Satterthwaite;
RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID; * Level 3;
RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID*TwinID; * Level 2;
ODS OUTPUT InfoCrit=Fit_RI2_RI3 CovParms=CovFQuad; * Save for LRT, pseudo-R2; RUN;
* Pseudo-R2 for time;
%PseudoR2(Ncov=3, CovFewer=CovEmpty, CovMore=CovFQuad);
```

* STATA Model 2a: Fixed Quadratic Time, Random Intercepts for Pair and Twin

```
mixed info c.time c.time#c.time ,
|| PairID: , covariance(unstructured) ///
|| TwinID: , covariance(unstructured) variance reml ///
dfmethod(satterthwaite) dftable(pvalue),
estimates store RI2_RI3
```

SAS output:

Cov Parm	Subject	Covariance Parameter Estimates		Z	Pr > Z
		Estimate	Standard Error		
UN(1,1)	PairID	88.0484	10.1556	8.67	<.0001
UN(1,1)	PairID*TwinID	52.9334	5.5159	9.60	<.0001
Residual		21.9701	0.9854	22.30	<.0001

The level-1 fixed linear and quadratic effects of time explained 8.33% of the level-1 residual variance. The level-2 twin intercept variance increased as a consequence.

		Information Criteria				
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
11211.6	3	11217.6	11217.6	11222.2	11229.2	11232.2
Solution for Fixed Effects						
Effect	Estimate	Standard Error	DF	t Value	Pr > t	
Intercept	26.1212	0.6233	369	41.91	<.0001	
time	-0.3216	0.1834	1040	-1.75	0.0797	
time*time	-0.03673	0.03077	1027	-1.19	0.2329	

PseudoR2 (% Reduction) for CovEmpty vs. CovFQuad

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovEmpty	UN(1,1)	PairID	87.2970	9.9794	8.75	<.0001	.
CovEmpty	UN(1,1)	PairID*TwinID	49.9360	5.3371	9.36	<.0001	.
CovEmpty	Residual		23.9684	1.0735	22.33	<.0001	.
CovFQuad	UN(1,1)	PairID	88.0484	10.1556	8.67	<.0001	-0.008607
CovFQuad	UN(1,1)	PairID*TwinID	52.9334	5.5159	9.60	<.0001	-0.060025
CovFQuad	Residual		21.9701	0.9854	22.30	<.0001	0.083373

Model 2b: Fixed Quadratic Time, Random Linear Time Slope at Level 2

Level 1: $Info_{ij} = \beta_{0ij} + \beta_{1ij}(Age_{ij} - PairAge1_j) + \beta_{2ij}(Age_{ij} - PairAge1_j)^2 + e_{ij}$

Level 2:

Intercept: $\beta_{0ij} = \delta_{00j} + U_{0ij}$

Linear Time: $\beta_{1ij} = \delta_{10j} + U_{1ij}$ ←

Quadratic Time: $\beta_{2ij} = \delta_{20j}$

Level 3:

Intercept: $\delta_{00j} = \gamma_{000} + V_{00j}$

Linear Time: $\delta_{10j} = \gamma_{100}$

Quadratic Time: $\delta_{20j} = \gamma_{200}$

```

TITLE "SAS Model 2b: Add Random Linear Time for Twin";
PROC MIXED DATA=work.Example8a NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS PairID TwinID;
  MODEL info = time time*time / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID; * Level 3;
  RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID*TwinID; * Level 2;
  ODS OUTPUT InfoCrit=Fit_RL2_RI3; * Save for LRT, pseudo-R2; RUN;
* Test random linear time at level 2;
%FitTest(FitFewer=Fit_RI2_RI3, FitMore=Fit_RL2_RI3);

* STATA Model 2b: Add Random Linear Time for Twin
mixed info c.time c.time#c.time ,
  || PairID: , covariance(unstructured) ///
  || TwinID: time , covariance(unstructured) variance reml ///
  dfmethod(satterthwaite) dftable(pvalue),
estimates store RL2_RI3
lrtest RI2_RI3 RL2_RL3

```

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	PairID	85.7639	9.7835	8.77	<.0001 → level-3 intercept var
UN(1,1)	PairID*TwinID	47.6649	5.2082	9.15	<.0001 → level-2 intercept var
UN(2,1)	PairID*TwinID	1.6668	0.8848	1.88	0.0596 → level-2 int-linear cov
UN(2,2)	PairID*TwinID	1.5662	0.2151	7.28	<.0001 → level-2 linear time var
Residual		13.5083	0.8175	16.52	<.0001 → level-1 residual var

Neg2LogLike	Parms	Information Criteria				
		AIC	AICC	HQIC	BIC	CAIC
11075.1	5	11085.1	11085.1	11092.7	11104.4	11109.4

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	26.1799	0.5991	338	43.70	<.0001
time	-0.3147	0.1583	929	-1.99	0.0471
time*time	-0.07075	0.02571	722	-2.75	0.0061

Likelihood Ratio Test for Fit_RI2_RI3 vs. Fit_RL2_RI3

Name	Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
Fit_RI2_RI3	11211.6	3	11217.6	11229.2	.	.	.
Fit_RL2_RI3	11075.1	5	11085.1	11104.4	136.518	2	0

Do we need the random linear time slope for twin?
Yes, $-2\Delta LL(\sim 2) = 136.52, p < .001$

Model 2c: Fixed Quadratic, Random Linear Slope at Levels 2 and 3

Level 1: $\text{Info}_{ij} = \beta_{0ij} + \beta_{1ij}(\text{Age}_{ij} - \text{PairAge1}_j) + \beta_{2ij}(\text{Age}_{ij} - \text{PairAge1}_j)^2 + e_{ij}$

Level 2:

Intercept: $\beta_{0ij} = \delta_{00j} + U_{0ij}$ Linear Time: $\beta_{1ij} = \delta_{10j} + U_{1ij}$ Quadratic Time: $\beta_{2ij} = \delta_{20j}$

Level 3:

Intercept: $\delta_{00j} = \gamma_{000} + V_{00j}$ Linear Time: $\delta_{10j} = \gamma_{100} + V_{10j}$ Quadratic Time: $\delta_{20j} = \gamma_{200}$

```
TITLE "SAS Model 2c: Add Random Linear Time for Pair";
PROC MIXED DATA=work.Example8a NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS PairID TwinID;
  MODEL info = time time*time / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID;          * Level 3;
  RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID*TwinID; * Level 2;
  ODS OUTPUT InfoCrit=Fit_RL2_RL3; * Save for LRT, pseudo-R2; RUN;
* Test random linear time at level 3;
%FitTest(FitFewer=Fit_RL2_RI3, FitMore=Fit_RL2_RL3);

* STATA Model 2c: Add Random Linear Time for Pair
mixed info c.time c.time#c.time ,
  || PairID: time, covariance(unstructured) ///
  || TwinID: time, covariance(unstructured) variance reml ///
  dfmethod(satterthwaite) dftable(pvalue),
  estimates store RL2_RL3
  lrtest RL2_RL3 RL2_RI3
```

SAS output:

ICC_{L3} for correlation of twins within pairs:
 For Intercept = 85.49 / (85.49 + 47.80) = .64
 For Linear Time = 0.11 / (0.11 + 1.45) = .07 (≈ 0)

Covariance Parameter Estimates		Estimate	Standard Error	Z	Pr > Z
Cov Parm	Subject				
UN(1,1)	PairID	85.4911	9.8263	8.70	<.0001 → level-3 intercept var
UN(2,1)	PairID	0.2432	1.0615	0.23	0.8188 → level-3 int-linear cov
UN(2,2)	PairID	0.1066	0.2203	0.48	0.3143 → level-3 linear time var
UN(1,1)	PairID*TwinID	47.7968	5.2453	9.11	<.0001 → level-2 intercept var
UN(2,1)	PairID*TwinID	1.5559	0.9849	1.58	0.1142 → level-2 int-linear cov
UN(2,2)	PairID*TwinID	1.4534	0.3050	4.77	<.0001 → level-2 linear time var
Residual		13.5251	0.8191	16.51	<.0001 → level-1 residual var

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
11074.8	7	11088.8	11088.8	11099.5	11115.8	11122.8

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	26.1810	0.5987	336	43.73	<.0001
time	-0.3181	0.1589	860	-2.00	0.0455
time*time	-0.07055	0.02573	721	-2.74	0.0062

Do we need the random linear slope for pair, too?
 Nope, $-2\Delta LL(\sim 2) = 0.29, p = .86$

Likelihood Ratio Test for Fit_RL2_RI3 vs. Fit_RL2_RL3							
Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
Fit_RL2_RI3	11075.1	5	11085.1	11104.4	.	.	.
Fit_RL2_RL3	11074.8	7	11088.8	11115.8	0.29080	2	0.86468

I then tested random quadratic time slopes at the twin and pair levels, but neither was significant. Given our interest in examining heritability of intercept and time slopes, we will retain the nonsignificant random linear time slope at level 3 (pairs) for now. So we continue by adding level-3 baseline age as a predictor of intercept and linear slope differences.

Model 3a: Add Baseline Age as a Predictor of Pair-Level Intercept and Time Slope Differences

Level 1: $Info_{tij} = \beta_{0ij} + \beta_{1ij}(Age_{tij} - PairAge1_j) + \beta_{2ij}(Age_{tij} - PairAge1_j)^2 + e_{tij}$

Level 2:

Intercept: $\beta_{0ij} = \delta_{00j} + U_{0ij}$

Linear Time: $\beta_{1ij} = \delta_{10j} + U_{1ij}$

Quadratic Time: $\beta_{2ij} = \delta_{20j}$

Level 3:

Intercept: $\delta_{00j} = \gamma_{000} + \gamma_{001}(PairAge1_j - 85) + V_{00j}$

Linear Time: $\delta_{10j} = \gamma_{100} + \gamma_{101}(PairAge1_j - 85) + V_{10j}$

Quadratic Time: $\delta_{20j} = \gamma_{200}$

```
TITLE "SAS Model 3a: Add Baseline Age as Predictor of Pair Intercepts and Linear Time Slopes";
PROC MIXED DATA=work.Example8a NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS PairID TwinID;
  MODEL info = time*time BFace85 time*BFace85 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID; * Level 3;
  RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID*TwinID; * Level 2;
  CONTRAST "Trajectory Diffs by Age" BFace85 1, time*Bface85 1 / CHISQ;
  ODS OUTPUT InfoCrit=Fit_Age CovParms=Cov_Age; * Save for LRT, pseudo-R2; RUN;
* Pseudo-R2 for age; %PseudoR2(Ncov=7, CovFewer=Cov_RL2_RL3, CovMore=Cov_Age);
```

```
* STATA Model 3a: Add Baseline Age as Predictor of Pair Intercepts and Linear Time Slopes
mixed info c.time c.time#c.time c.BFage85 c.time#c.BFage85, ///
  || PairID: time , covariance(unstructured) ///
  || TwinID: time , covariance(unstructured) variance reml ///
  dfmethod(satterthwaite) dftable(pvalue),
test (c.BFage85=0) (c.time#c.BFage85=0) // Trajectory diffs by age
```

SAS output:

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	PairID	78.7908	9.3017	8.47	<.0001
UN(2,1)	PairID	-0.02415	1.0154	-0.02	0.9810
UN(2,2)	PairID	0.07234	0.2193	0.33	0.3707
UN(1,1)	PairID*TwinID	47.6089	5.2158	9.13	<.0001
UN(2,1)	PairID*TwinID	1.6686	0.9812	1.70	0.0890
UN(2,2)	PairID*TwinID	1.4534	0.3052	4.76	<.0001
Residual		13.5712	0.8236	16.48	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
11056.3	7	11070.3	11070.4	11081.1	11097.4	11104.4

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	24.8887	0.6473	345	38.45	<.0001
time	-0.4284	0.1681	892	-2.55	0.0110
time*time	-0.07124	0.02580	717	-2.76	0.0059
BFage85	-0.8602	0.1864	348	-4.61	<.0001
time*BFage85	-0.05655	0.03089	267	-1.83	0.0683

The level-3 main effect of age and its interaction with time explained 7.84% and 32.11% of the level-3 pair intercept and time slope variance, respectively. I also tried quadratic effects of age in predicting the intercept and linear time slope, but neither was significant.

Contrasts						
Label	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
Trajectory Diffs by Age	2	302	25.80	12.90	<.0001	<.0001

PseudoR2 (% Reduction) for Cov_RL2_RL3 vs. Cov_Age

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
Cov_RL2_RL3	UN(1,1)	PairID	85.4911	9.8263	8.70	<.0001	.
Cov_RL2_RL3	UN(2,2)	PairID	0.1066	0.2203	0.48	0.3143	.
Cov_RL2_RL3	UN(1,1)	PairID*TwinID	47.7968	5.2453	9.11	<.0001	.
Cov_RL2_RL3	UN(2,2)	PairID*TwinID	1.4534	0.3050	4.77	<.0001	.
Cov_RL2_RL3	Residual		13.5251	0.8191	16.51	<.0001	.
Cov_Age	UN(1,1)	PairID	78.7908	9.3017	8.47	<.0001	0.07837
Cov_Age	UN(2,2)	PairID	0.07234	0.2193	0.33	0.3707	0.32109
Cov_Age	UN(1,1)	PairID*TwinID	47.6089	5.2158	9.13	<.0001	0.00393
Cov_Age	UN(2,2)	PairID*TwinID	1.4534	0.3052	4.76	<.0001	-0.00003
Cov_Age	Residual		13.5712	0.8236	16.48	<.0001	-0.00340

Model 3b: Add Zygosity as a Predictor of Pair-Level Intercept and Time Slope Differences

Level 1: $Info_{ij} = \beta_{0ij} + \beta_{1ij}(Age_{ij} - PairAge1_j) + \beta_{2ij}(Age_{ij} - PairAge1_j)^2 + e_{ij}$

Level 2:

Intercept: $\beta_{0ij} = \delta_{00j} + U_{0ij}$

Linear Time: $\beta_{1ij} = \delta_{10j} + U_{1ij}$

Quadratic Time: $\beta_{2ij} = \delta_{20j}$

Level 3:

Intercept: $\delta_{00j} = \gamma_{000} + \gamma_{001}(PairAge1_j - 85) + \gamma_{002}(MZvDZ_j) + \gamma_{003}(PairAge1_j - 85)(MZvDZ_j) + V_{00j}$

Linear Time: $\delta_{10j} = \gamma_{100} + \gamma_{101}(PairAge1_j - 85) + \gamma_{102}(MZvDZ_j) + \gamma_{103}(PairAge1_j - 85)(MZvDZ_j) + V_{10j}$

Quadratic Time: $\delta_{20j} = \gamma_{200}$

```
TITLE "SAS Model 3b: Add Zygosity as Predictor of Pair Intercepts and Linear Time Slopes";
PROC MIXED DATA=work.Example8a NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
  CLASS PairID TwinID zyg;
  MODEL info = time*time BFace85 time*BFace85
           zyg zyg*time zyg*BFace85 zyg*time*BFace85 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID;          * Level 3;
  RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID*TwinID;   * Level 2;
  CONTRAST "Diffs by Zyg" zyg -1 1, time*zyg -1 1, BFace85*zyg -1 1, time*BFace85*zyg -1 1 / CHISQ;
  ODS OUTPUT InfoCrit=Fit_Zyg CovParms=Cov_Zyg; * Save for LRT, pseudo-R2; RUN;
  * Pseudo-R2 for zygosity;
  %PseudoR2(Ncov=7, CovFewer=Cov_Age, CovMore=Cov_Zyg);

* STATA Model 3b: Add Zygosity as Predictor of Pair Intercepts and Linear Time Slopes
mixed info c.time c.time#c.time c.BFace85 c.time#c.BFace85 ///
      c.zyg c.zyg#c.time c.zyg#c.BFace85 c.zyg#c.time#c.BFace85, ///
  || PairID: time , covariance(unstructured) ///
  || TwinID: time , covariance(unstructured) variance reml ///
      dfmethod(satterthwaite) dftable(pvalue),
  // Trajectory diffs by zygosity
  test (c.zyg=0) (c.zyg#c.time=0) (c.zyg#c.BFace85=0) (c.zyg#c.time#c.BFace85=0)
  estimates store Fit_Zyg
```

SAS output:

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	PairID	76.9815	9.2255	8.34	<.0001
UN(2,1)	PairID	0.1952	1.0214	0.19	0.8484
UN(2,2)	PairID	0.07385	0.2177	0.34	0.3672
UN(1,1)	PairID*TwinID	47.8176	5.2339	9.14	<.0001
UN(2,1)	PairID*TwinID	1.6538	0.9833	1.68	0.0926
UN(2,2)	PairID*TwinID	1.4464	0.3021	4.79	<.0001
Residual		13.5287	0.8181	16.54	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
11048.7	7	11062.7	11062.7	11073.4	11089.7	11096.7

Contrasts							
Label	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F	
Trajectory Diffs by Zygosity	4	276	11.30	2.83	0.0234	0.0253	

Solution for Fixed Effects

Effect	zyg	Estimate	Standard Error	DF	t Value	Pr > t
Intercept		26.2390	0.9772	327	26.85	<.0001
time		-0.3668	0.2039	646	-1.80	0.0724
time*time		-0.07171	0.02577	720	-2.78	0.0055
BFace85		-1.0161	0.2820	328	-3.60	0.0004
time*BFace85		0.01414	0.04557	212	0.31	0.7566
zyg	DZ	-2.3236	1.2924	333	-1.80	0.0731
zyg	MZ	0
time*zyg	DZ	-0.1225	0.2061	262	-0.59	0.5529
time*zyg	MZ	0
BFace85*zyg	DZ	0.2774	0.3737	341	0.74	0.4584
BFace85*zyg	MZ	0
time*BFace85*zyg	DZ	-0.1308	0.06181	257	-2.12	0.0352
time*BFace85*zyg	MZ	0

The level-3 main effect of zygosity explained 2.61% of the level-3 pair intercept variance, but zygosity by time actually increased the level-3 pair slope variance instead.

PseudoR2 (% Reduction) for Cov_Age vs. Cov_Zyg

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
Cov_Age	UN(1,1)	PairID	78.7908	9.3017	8.47	<.0001	.
Cov_Age	UN(2,2)	PairID	0.07234	0.2193	0.33	0.3707	.
Cov_Age	UN(1,1)	PairID*TwinID	47.6089	5.2158	9.13	<.0001	.
Cov_Age	UN(2,2)	PairID*TwinID	1.4534	0.3052	4.76	<.0001	.
Cov_Age	Residual		13.5712	0.8236	16.48	<.0001	.
Cov_Zyg	UN(1,1)	PairID	76.7361	9.2038	8.34	<.0001	0.026078
Cov_Zyg	UN(2,2)	PairID	0.07387	0.2196	0.34	0.3683	-0.021145
Cov_Zyg	UN(1,1)	PairID*TwinID	47.8637	5.2448	9.13	<.0001	-0.005352
Cov_Zyg	UN(2,2)	PairID*TwinID	1.4538	0.3048	4.77	<.0001	-0.000289
Cov_Zyg	Residual		13.5682	0.8232	16.48	<.0001	0.000220

Model 3c: Add Heterogeneous Variances by Zygosity (to quantify heritability)

Note: The STATA version required creating extra dummy codes for the MZ and DZ main effects and interactions with time to be used in the variance model.

```

TITLE "SAS Model 3c: Add Heterogeneous G and R matrices by Zygosity";
PROC MIXED DATA=work.Example8a NOCLPRINT COVTEST NAMELEN=100 IC METHOD=REML;
CLASS PairID TwinID zyg;
MODEL info = time time*time BFace85 time*BFace85
zyg zyg*time zyg*BFace85 zyg*time*BFace85 / SOLUTION DDFM=Satterthwaite;
RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID GROUP=zyg; * Level 3;
RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID*TwinID GROUP=zyg; * Level 2;
REPEATED / GROUP=zyg;
ODS OUTPUT InfoCrit=Fit_Het CovParms=Cov_Het; * Save for LRT, pseudo-R2;
ESTIMATE "Age on Intercept: DZ" BFace85 1 BFace85*zyg 1 0;
ESTIMATE "Age on Time Slope: DZ" time*BFace85 1 time*BFace85*zyg 1 0; RUN;
* Test het variances;
%FitTest(FitFewer=Fit_Zyg, FitMore=Fit_Het);

* STATA Model 3c: Add Heterogeneous G and R matrices by Zygosity
mixed info c.time c.time#c.time c.BFace85 c.time#c.BFace85 ///
c.zyg c.zyg#c.time c.zyg#c.BFace85 c.zyg#c.time#c.BFace85, ///
|| PairID: mz mzttime , noconstant covariance(unstructured) ///
|| PairID: dz dztime , noconstant covariance(unstructured) ///
|| TwinID: mz mzttime , noconstant covariance(unstructured) ///
|| TwinID: dz dztime , noconstant covariance(unstructured) ///
variance reml dfmethod(satterthwaite) dftable(pvalue) residuals(independent,by(zyg))
lincom c.BFace85*1 + c.zyg#c.BFace85*1 // Age on Intercept: DZ
lincom c.time#c.BFace85*1 + c.zyg#c.time#c.BFace85*1 // Age on Time Slope: DZ
estimates store Fit_Het
lrtest Fit_Het Fit_Zyg
    
```

SAS output:

Covariance Parameter Estimates						
Cov Parm	Subject	Group	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	PairID	zyg DZ	55.0442	11.8158	4.66	<.0001
UN(2,1)	PairID	zyg DZ	-0.4171	1.3047	-0.32	0.7492
UN(2,2)	PairID	zyg DZ	0	.	.	.
UN(1,1)	PairID	zyg MZ	105.88	15.0698	7.03	<.0001
UN(2,1)	PairID	zyg MZ	0.9788	1.7090	0.57	0.5668
UN(2,2)	PairID	zyg MZ	0.6152	0.3648	1.69	0.0459
UN(1,1)	PairID*TwinID	zyg DZ	70.8603	9.5620	7.41	<.0001
UN(2,1)	PairID*TwinID	zyg DZ	2.4174	1.3398	1.80	0.0712
UN(2,2)	PairID*TwinID	zyg DZ	1.1609	0.2462	4.71	<.0001
UN(1,1)	PairID*TwinID	zyg MZ	18.5869	4.0696	4.57	<.0001
UN(2,1)	PairID*TwinID	zyg MZ	0.4866	1.0519	0.46	0.6436
UN(2,2)	PairID*TwinID	zyg MZ	1.3806	0.4153	3.32	0.0004
Residual		zyg DZ	13.9688	1.1309	12.35	<.0001
Residual		zyg MZ	12.9889	1.1721	11.08	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
11005.0	13	11031.0	11031.2	11051.0	11081.2	11094.2

Solution for Fixed Effects						
Effect	zyg	Estimate	Standard Error	DF	t Value	Pr > t
Intercept		26.1032	1.0277	139	25.40	<.0001
time		-0.3426	0.2169	289	-1.58	0.1154
time*time		-0.07051	0.02570	722	-2.74	0.0062
BFace85		-1.0285	0.2963	139	-3.47	0.0007
time*BFace85		0.03232	0.05169	102	0.63	0.5332
zyg	DZ	-2.1640	1.3125	289	-1.65	0.1003
zyg	MZ	0
time*zyg	DZ	-0.1410	0.2154	228	-0.65	0.5135
time*zyg	MZ	0
BFace85*zyg	DZ	0.2888	0.3799	295	0.76	0.4477
BFace85*zyg	MZ	0
time*BFace85*zyg	DZ	-0.1515	0.06481	221	-2.34	0.0203
time*BFace85*zyg	MZ	0

Estimates						
Label	Estimate	Standard Error	DF	t Value	Pr > t	
Age on Intercept: DZ	-0.7397	0.2378	212	-3.11	0.0021	
Age on Time Slope: DZ	-0.1192	0.03919	236	-3.04	0.0026	

Likelihood Ratio Test for Fit_Zyg vs. Fit_Het

Is the heterogeneous variance model a better fit?
 Yes, $-2\Delta LL(7) = 43.66$, $p < .001$ (note SAS didn't count the 0)

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
Fit_Zyg	11048.7	7	11062.7	11089.7	.	.	.
Fit_Het	11005.0	13	11031.0	11081.2	43.6599	6	8.6337E-8

Heritability (A or H^2), or the contribution of genetics, can be found as twice the difference of the intraclass correlation (ICC) between MZ and DZ twins. **Common environment** (C^2) can be found as the difference between the ICC for MZ twins and the heritability estimate (usually constrained to be ≥ 0), and the **unique environment** (E^2) can be found as the remainder (i.e., $1 - [\text{heritability} + \text{common environment}]$). Applying these calculations to our results reveals evidence for heritability in both the intercept and the linear time slope, but with much greater uncertainty in the latter.

Intercept	Intercept			Linear Time Slope		
	DZ	MZ	HCE	DZ	MZ	HCE
Level-3 Pair Variance	55.044	105.880		0.000	0.615	
Level-2 Twin Variance	70.860	18.587		1.161	1.381	
ICC = $L3 / (L3 + L2)$	0.437	0.851		0.000	0.308	
$H^2 = 2 * (\text{ICC MZ} - \text{ICC DZ})$			0.827			0.616
$C^2 = \text{ICC MZ} - H^2$			0.024			-0.308
$E^2 = 1 - (H^2 + C^2)$			0.149			0.692

Sample Results Section:

The extent of individual change in crystallized intelligence (as measured by the information test) and the extent of heritability therein was examined in a sample of 351 same-sex twin pairs measured every two years for up to four occasions. Multilevel models were estimated using residual maximum likelihood in SAS MIXED. Accordingly, the significance of fixed effects was evaluated with Wald tests using Satterthwaite denominator degrees of freedom, whereas the significance of random effects was evaluated via likelihood ratio tests (i.e., $-2\Delta LL$ with degrees of freedom equal to the number of new random effects variances and covariances). Pseudo- R^2 effect sizes for the fixed effects were calculated as the proportion reduction in each variance component.

A two-level empty means, random intercept model of occasions at level 1 nested in persons at level 2 was initially estimated; its intraclass correlation (ICC) indicated that 85% of the outcome variance was between persons. The addition of a level-3 random intercept for twin pair resulted in significantly better model fit, $-2\Delta LL(1) = 111.37$, $p < .001$, and revealed that, of the 85% of the outcome variance that was between persons, 64% was actually due to twin pair (i.e., shared variance between twins). Stated more directly, of the total variance, 15% was at level 1 (within persons over time), 31% was at level 2 (between twins), and 54% was at level 3 (between pairs). A three-level empty means, random intercept model to partition the variance in time-varying age revealed that 63% was between pairs (given that the twins varied in age from 80 to 100 at baseline), whereas the remaining 37% was within persons over time; there was no level-2 age variance in these twins. Thus, the level-3 (cross-sectional) and level-1 (longitudinal) effects of age were modeled separately using baseline age (centered so $0 = 85$) and time in study (with $0 = \text{baseline}$), respectively.

Based on the pattern of model-estimated means, fixed linear and quadratic effects of time were first added, which accounted for 8.33% of the level-1 residual variance. Although adding a variance for the level-2 (twin) random linear time slope (and its covariance with the level-2 twin intercept) significantly improved model fit, $-2\Delta LL(2) = 136.52$, $p < .001$, the subsequent addition of a variance for the level-3 (pair) random linear time slope (and its covariance with the level-3 pair intercept) did not significantly improve model fit, $-2\Delta LL(2) = 0.29$, $p = .86$, indicating that the 7% of the random linear time slope variance that was due to twin pair was not distinguishable from 0. Given our interest in examining heritability, though, both random linear time slope variances were retained. Random quadratic time slopes were not significant at either level 2 or level 3, and these were not retained.

The effect of baseline age on the intercept and linear time slope was then added, which explained 7.84% and 32.11% of the level-3 intercept and linear time slope variance, respectively, and which resulted in significant model improvement, $F(2,302) = 12.90$, $p < .001$. We then added zygosity mean differences in the intercept, linear time slope, and the effects of baseline age on the intercept and linear time slope. Although these four new fixed effects also resulted in significant model improvement, $F(4,276) = 2.83$, $p < .001$, only the level-3 pair intercept variance was reduced (by 2.61%); the level-3 pair time slope variance increased by 2.11% instead. Finally, we added zygosity differences in all variance model parameters—three at level 3, three at level 2, and in residual variance at level 1, which resulted in significant model improvement, $-2\Delta LL(7) = 43.66$, $p < .001$.

Results for the final model are given in Table X. Given the centering of the model predictors, the reference for the intercept and linear time slope is an MZ twin pair who were 85 years at baseline (when time = 0). Older age at baseline was related to a significantly lower intercept at time 0, equivalently so in both MZ and DZ twins. In contrast, the interaction of age by linear time differed significantly by zygosity: older age at baseline was related to a significantly more negative linear time slope in DZ twins, but not in MZ twins (in which the interaction of age by time was nonsignificantly positive instead). There was also a significant fixed quadratic effect of time, which indicated that the linear rate of decline became more negative by twice the quadratic coefficient of 0.07 per year (i.e., steeper longitudinal decline later in the study, unconditional by baseline age or zygosity). (see text above for interpretation of heritability results)